<commit_before>prefix bag: <http://bag.basisregistraties.overheid.nl/def/bag#>
prefix geo: <http://www.opengis.net/ont/geosparql#>
prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select ?wkt ?wktLabel {
  bind ("Warnsveld" as ?woonplaatsNaam)
  bind ("'t Spiker" as ?straatNaam)
  ?openbareRuimte bag:naamOpenbareRuimte ?straatNaam ;
                  bag:bijbehorendeWoonplaats ?woonplaats .
  ?woonplaats bag:naamWoonplaats ?woonplaatsNaam .
  ?nummeraanduiding bag:bijbehorendeOpenbareRuimte ?openbareRuimte ;
                    bag:huisnummer ?huisnummer .
  ?verblijfsobject bag:hoofdadres ?nummeraanduiding ;
                   bag:oppervlakte ?oppervlakte ;
                   bag:pandrelatering ?pand ;
                   bag:status ?verblijfsobjectStatus .
  ?verblijfsobjectStatus rdfs:label ?verblijfsobjectStatusLabel .
  ?pand bag:oorspronkelijkBouwjaar ?bouwjaar ;
        bag:geometriePand/geo:asWKT ?wkt ;
        bag:status ?pandStatus .
  ?pandStatus rdfs:label ?pandStatusLabel
  bind (concat('<h4>',str(?straatNaam),' ',str(?huisnummer),'</h4><h5><a href="',str(?verblijfsobject),'">Verblijfsobject</a>:</h5><ul><li>Woonplaats: <a href="',str(?woonplaats),'">',str(?woonplaatsNaam),'</a></li><li>Oppervlakte: ',str(?oppervlakte),'m²</li><li>Status: <a href="',str(?verblijfsobjectStatus),'">',str(?verblijfsobjectStatusLabel),'</a></li></ul><h5><a href="',str(?pand),'">Pand</a>:</h5><ul><li>Bouwjaar: ',str(?bouwjaar),'</li><li>Status: <a href="',str(?pandStatus),'">',str(?pandStatusLabel),'</a></li></ul>') as ?wktLabel)
}<commit_msg>ADDED: energy labels to the energy street query.<commit_after>prefix bag: <http://bag.basisregistraties.overheid.nl/def/bag#>
prefix energie: <http://data.labs.pdok.nl/dataset/energie#>
prefix geo: <http://www.opengis.net/ont/geosparql#>
prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select ?wkt ?wktColor ?wktLabel {
  bind ("Warnsveld" as ?woonplaatsNaam)
  bind ("'t Spiker" as ?straatNaam)
  ?openbareRuimte bag:naamOpenbareRuimte ?straatNaam ;
                  bag:bijbehorendeWoonplaats ?woonplaats .
  ?woonplaats bag:naamWoonplaats ?woonplaatsNaam .
  ?nummeraanduiding bag:bijbehorendeOpenbareRuimte ?openbareRuimte ;
                    bag:huisnummer ?huisnummer ;
                    bag:postcode ?postcode .
  ?verblijfsobject bag:hoofdadres ?nummeraanduiding ;
                   bag:oppervlakte ?oppervlakte ;
                   bag:pandrelatering ?pand ;
                   bag:status ?verblijfsobjectStatus .
  ?verblijfsobjectStatus rdfs:label ?verblijfsobjectStatusLabel .
  ?pand bag:oorspronkelijkBouwjaar ?bouwjaar ;
        bag:geometriePand/geo:asWKT ?wkt ;
        bag:status ?pandStatus .
  ?pandStatus rdfs:label ?pandStatusLabel .
  bind (str(?huisnummer) as ?huisnummerString)
  service <https://api.krr.triply.cc/datasets/Kadaster/geosoup2/services/geosoup/sparql> {
    optional {
      ?x energie:ep ?ep ;
         energie:e_label ?energielabel ;
         energie:opname_datum ?opnamedatum ;
         energie:pand_huisnummer ?huisnummerString ;
         energie:pand_postcode ?postcode ;
         energie:registratie_datum ?registratiedatum .
    }
  }
  bind (if(?energielabel="A","rgb(34,177,76)",
        if(?energielabel="B","rgb(49,68,32)",
        if(?energielabel="C","rgb(76,90,43)",
        if(?energielabel="D","rgb(90,87,43)",
        if(?energielabel="E","rgb(80,59,39)",
        if(?energielabel="F","rgb(70,47,32)",
        if(?energielabel="G","rgb(102,51,53)",
        ?grey))))))) as ?wktColor)
  bind (concat('<h4>',str(?straatNaam),' ',str(?huisnummer),'</h4>',if(bound(?x),concat('<h5>Energielabel informatie:</h5><ul><li>Energielabel: ',str(?energielabel),'</li><li>EP: ',str(?ep),'</li><li>Opnamedatum: ',str(?opnamedatum),'</li><li>Registratiedatum: ',str(?registratiedatum),'</li></ul>'),''),'<h5><a href="',str(?verblijfsobject),'">Verblijfsobject</a>:</h5><ul><li>Woonplaats: <a href="',str(?woonplaats),'">',str(?woonplaatsNaam),'</a></li><li>Oppervlakte: ',str(?oppervlakte),'m²</li><li>Status: <a href="',str(?verblijfsobjectStatus),'">',str(?verblijfsobjectStatusLabel),'</a></li></ul><h5><a href="',str(?pand),'">Pand</a>:</h5><ul><li>Bouwjaar: ',str(?bouwjaar),'</li><li>Status: <a href="',str(?pandStatus),'">',str(?pandStatusLabel),'</a></li></ul>') as ?wktLabel)
}
{'fraction_non_alphanumeric': 0.19246861924686193, 'fraction_numerical': 0.01696885169688517, 'mean_word_length': 4.737333333333333, 'pattern_counts': {'":': 0, '<': 83, '<?xml version=': 0, '>': 83, 'https://': 1, 'lorem ipsum': 0, 'www.': 4, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '96144', 'n_tokens_mistral': 1816, 'n_tokens_neox': 1757, 'n_words': 191}